

Julia for high-performance simulation of optical pulses

Corentin Simon

Univesité Libre de Bruxelles

2026-02-10

Simulations of optical pulse

In general, we want to integrate an equation of the form

$$\partial_z u(z, t) = \sum_n a_n \partial_t^n u + \hat{N}(u)$$

\Rightarrow Semilinear parabolic PDE

$$\partial_z u(z, t) = \hat{D}u + \hat{N}(u)$$

- \hat{D} is a (stiff-linear) operator.
- \hat{N} is some nonlinear function on u .

Stiffness problem

\hat{D} is stiff : **large eigenvalues**.

If transformed in the spectral domain $\partial_t \rightarrow -i\omega$

$$\hat{D}(\omega) = \sum_n a_n (-i\omega)^n$$

- Standard explicit ODE Method (RK45, ...) poorly suited due to stiffness.
- Implicit methods are better but required a nonlinear solver
 - Large system ($10^2 - 10^3$ or more points) \Rightarrow **poor performance** !

Split-step method

Split-step method

For the NLSE equation:

$$\partial_z u(z, t) = i \frac{\beta_2}{2} \partial_t^2 u + i \gamma |u|^2 u$$

Exact solution

- $u(L, t) = \exp\left(L\left(\hat{D} + \hat{N}\right)\right)u(0, t)$
- Cannot expand the exponential as \hat{D} and \hat{N} do not commute.
- Uses the BCH approximation

$$\exp\left(h\left(\hat{D} + \hat{N}\right)\right)u \approx \exp\left(h\hat{D}\right)\exp\left(h\hat{N}\right) + \mathcal{O}(h^2)$$

for small step h

- Split the step between linear and nonlinear one for a small
- Repeats L/h times.

Split-step method

Split-step implementation

For the NLSE,

- $\exp(h\hat{N})u_0 = \exp(i\gamma h |u_0|^2)u_0$

which can be directly computed

- For $\exp(h\hat{D})u_0$, \hat{D} is diagonal in the spectral representation

\Rightarrow exponentiation in the spectral domain

$\Rightarrow \exp(h\hat{D})u_0 = \mathcal{F}^{-1}(\exp(h\hat{D}(\omega))\mathcal{F}(u_0))$

Fourier transform for performing the DFT \mathcal{F} on u .

- Very performant $\mathcal{O}(n \log n)$
- Very fast/optimized implentation exists (MKL, FFTW, ...)

Split-step method

Split-step: strength and weakness

Strength

- Easy to implement

Weaknesses

- Lack accuracy $\mathcal{O}(h^2)$
- No adaptative control: fixed time step.
 - adaptative variations exist but are harder to implement.
- Required n exponential each step for the nonlinear part
 - Exponential is expensive
- For more complex nonlinear part \hat{N} , no explicit formula for $\exp(\hat{N})$
 - Numerical integration is required (RK45, ...)

RKIP - Stands for **R**unge-**K**utta in the **I**nteraction **P**icture

- Alternative method to split-step.
- Transform the stiff-problem into a non stiff one.
- Allow to use direct Runge-Kutta methods.

If we defined u_I as

$$\begin{aligned}
 u_I(z, t) &= \exp(-t\hat{D})u \\
 \partial_t u_I &= -\hat{D}u_I + \exp(-t\hat{D})\partial_t u \\
 &= -\hat{D}u_I + \exp(-t\hat{D})(\hat{D}u + \hat{N}(u)) \\
 &= -\hat{D}u_I + \hat{D}\exp(-t\hat{D})u + \exp(-t\hat{D})\hat{N}(u) \\
 &= \exp(-t\hat{D})\hat{N}(u) = \exp(-t\hat{D})\hat{N}(\exp(t\hat{D})u_I)
 \end{aligned}$$

For a Runge-Kutta tableau $c_i, a_{ij}, \alpha_i, \tilde{\alpha}_i$, this gives

$$k_i = \exp(-c_i h \hat{D}) \hat{N} \left(\exp(c_i h \hat{D}) \left(u_n + \sum_{j < i} h a_{ij} k_j \right) \right)$$

$$u_{n+1} = \exp(h \hat{D}) \left(u_n + \sum_i h \alpha_i k_i \right) \text{ (Update)}$$

$$e = \exp(h \hat{D}) \left(\sum_i h (\alpha_i - \tilde{\alpha}_i) k_i \right) \begin{array}{l} \text{(Error estimate} \\ \text{if adaptative)} \end{array}$$

The best tableau for this method seems to be the Verner 5(6) tableau.

RKIP

RKIP: strength and weakness

Strength

- As high order as the RK tableau
 - Large repertoire or higher-order efficient tables
- Adaptative with error control
- Agnostic about \hat{N}

Weaknesses

- Required computing $\exp(h\hat{D})$ each steps
 - If not adaptative, can be precomputed
 - If adaptative, can be cached for a fixed number of time steps h_1, h_2, \dots, h_S
→ restrict h to theses
- **Hard to implement**
 - Good new : **you don't have to !**

RKIP implentation

RKIP is now available as part of the Julia `OrdinaryDiffEq.jl` package.

Why Julia ?

- Julia is compiled Just-in-Time (JIT)
- Syntax close to Python and Matlab
- Faster interpreted language (even with tools/precompiled library)
- `OrdinaryDiffEq.jl` is the most performant ODE solver library available

How to use this code repo

How to use this code repo

1. Install Julia <https://julialang.org/downloads/> and git <https://git-scm.com/install/windows>

For Windows:

```
1 winget install --name Julia --id 9NJNW8PVKMN -e -s msstore
2 winget install --id Git.Git -e --source winget
```

[shell](#)

2. Ensure you have VSCode installed (<https://code.visualstudio.com/download>), with its Julia extension.
3. Clone the folder, and open it in VSCode.

```
1 git clone https://github.com/Azercoco/NLSE-Julia-Example.git
2 cd NLSE-Julia-Example
3 code .
```

[shell](#)

How to use this code repo

Once VSCode is open in the folder, open the command palette (Default : `Ctrl+Shift+P`)

Search for `> Julia` : Restart REPL, then execute with `Enter`

This will open the **Julia REPL** for interacting with Julia.

In the Julia REPL, type

```
1 ] # Open the Package manager
2 activate .
3 instantiate
```

`julia`

This will download and install all the required packages.

You can now play or use one of the examples.

RKIP Examples

Example 1 : LLE

See `../examples/example_1_lle.jl`

This example show how to solve the Lugiato-Levefer equation

$$\partial_T u(T, t) = (-1 + i\partial_t^2)u + i(\Delta - |u|^2)u + S$$

which has two parameters S and Δ .

The operators are:

- $\hat{D} = -1 + i\partial_t^2$
- $\hat{N}(u) = i(\Delta - |u|^2)u + S$

Δ could have also been moved into \hat{D} but it is easier to include in \hat{N} as we want to be able to vary it with time $\Delta(t)$.

RKIP Examples

Dependency loading.

```
1 import ProgressLogging
2 using Plots
3 include("../src/mod.jl")
```

julia

- ProgressLogging is used to display a progress bar in VSCode and Plots for plotting.
- src/mod.jl contains all the routine used for the examples.

In Julia, a differential equation is defined as $f(u, p, t)$ where t is the integration variable and p a `NamedTuple` containing the equation parameters (here S and Δ).

```
1 p = (; S=1.1, Δ=1.1);
```

julia

This code create a `NamedTuple` with field S and Δ .

RKIP Examples

Crating the initial conditions.

```
1 u0 = lle_hss(p.S, p.Δ, 2^12; noise=1e-2)
```

julia

This creates a vector of size $2^{12} = 2048$ points, corresponding to the continuous stationary solution of the LLE equation with a random noise level of noise.

```
1 dτ = 0.05
```

julia

```
2 LLE = SemilinearPDE(
```

```
3     StandartLLE(),
```

```
4     u0,
```

```
5     300.0,
```

```
6     dτ,
```

```
7     p
```

```
8 );
```

This creates a semilinear PDE with initial solution $u0$, an integration bounds of $(0, 300.0)$, a time step $d\tau$ and parameters p .

We solve the problem

```
1 u = solve_rkip(LLE; abstol=1e-4, reltol=1e-8)
```

julia

with relative tolerance `reltol` and absolute tolerance `abstol`. Changing these value will increase/decrease the precision with a cost/gain in performance.

We can plot the solution:

```
1 plot(  
2     LLE.τ,  
3     abs2.(u),  
4 )
```

julia

LLE has two fields `τ` and `freq` storing the value of fast time and normalized frequencies.

We can also sample the solution at different time using the `saveat` keyword arguments.

```
1 t = 0.0:5:300 # sample every 5 unit of time between 0 and 300
2 u2 = solve_rkip(LLE; saveat=t, abstol=1e-4, reltol=1e-8);
```

[julia](#)

`u2` is now matrix with a number of columns corresponding the size of `t`.

Similarly the solution is plotted.

```
1 heatmap(abs2.(u2))
```

[julia](#)

`LLE` has two fields `τ` and `freq` storing the value of fast time and normalized frequencies.

Example 2: LLE-Scan with time varying Δ

See `../examples/example_2_lle_scan.jl`

This example is similar to the first. The only difference is that Δ is now a function of p and t .

```

1   $\Delta_{\text{scan}}(p, t) = p.\Delta_0 + p.\Delta_{\text{rate}} * t$ 
2  p = (;
3      S=3.5,
4       $\Delta_0=-2.0$ ,
5       $\Delta_{\text{rate}}=0.2$ ,
6       $\Delta = \Delta_{\text{scan}}$ ,
7  )
8  u0 = lle_hss(p.S, p. $\Delta_0$ , 1024; noise=1e-2)
```

We have introduced two new parameters, the initial detuning Δ_0 and detuning scan rate Δ_{rate} . Otherwise, we solve it exactly the same manner as the first example.

Example 3: third-order dispersion

See `../examples/example_3_lle_d3.jl`

In this example, we want to modify the linear operator \hat{D} to add a third-order dispersion to the LLE:

$$\partial_T u(T, t) = (-1 + i\partial_t^2 + id_3\partial_t^3)u + \dots$$

We add this new parameters d_3 to the set of parameters.

```
1 p = (;..., d3=0.08)
```

julia

```
1 LLE_d3 = SemilinearPDE(
```

julia

```
2     StandartLLE(;
```

```
3          $\hat{D}=(\omega, p) \rightarrow -1 - 1im * \omega^2 + 1im * \omega^3 * p.d_3$ 
```

```
4     ),...
```

We can override the the definition of \hat{D} by passing a function of (ω, p) computing $\tilde{\hat{D}}(\omega)$.

RKIP Examples

Example 4: Raman Nonlinearity

See `../examples/example_4_lle_rarman.jl`

The Raman nonlinearity adds an additional term to the LLE:

$$\partial_T u(T, t) = (-1 + i\partial_t^2)u + i(\Delta - |u|^2)u + S + i\tau_R(\partial_t |u|^2)u$$

Similarly, we add a new parameter `tau_R`.

```
1 p = (;..., τ_R=2e-4)
```

julia

And we can modify the nonlinearity used by doing

```
1 LLE_raman = SemilinearPDE(
2     StandartLLE(;
3         Ñ=LLE_RamanNonlinearPart()
4     ),...
```

julia

The rest is similar to the previous examples.

Adding a custom nonlinearity.

To add a new nonlinearity, declares a new empty struct corresponding this nonlinearity, which inherits `AbstractNonlinearPart` and implements `(nl::CustomNonlinearity)(du, u, p, t)`.

```
1 struct CustomNonlinearity <: AbstractNonlinearPart end
2
3 function (nl::CustomNonlinearity)(du, u, p, t)
4     # Put your nonlinear function here and store the result in du
5 end
```

`julia`

and then use:

```
1 custom_lle = SemilinearPDE(
2     StandartLLE(;
3         N=CustomNonlinearity()
4     ),...
```

`julia`

Custom Nonlinearity

Custom Nonlinearity: PDLNSE

Example nonlinearity for a parametric driven NLSE:

$$\partial_T u(T, t) = (-1 + i\partial_t^2)u + i(\Delta - |u|^2)u + \kappa u^*$$

```

1 struct PDLNSE <: AbstractNonlinearPart end
2
3 function (nl::PDLNSE)(du, u, p, t)
4     Δ = get_var(nl, p.Δ, p, t)
5     κ = get_var(nl, p.κ, p, t)
6
7     @. du = lim * (abs2(u) - Δ) * u + κ * conj(u)
8 end

```

julia

The helper `get_var(nl, var, p, t)` allows `var` to be either a scalar or a function of `(p, t)`. We used the *broadcast* macro `@.` which executes the operation on the array without allocations.

Also remember to add `κ=...` to `p`.

Custom Nonlinearity

Custom Nonlinearity: Cache

Sometimes we want to cache some array for the computation. In that case, we can implement function `get_cache(nl, u0, p)` which returns a `NamedTuple`.

```
1 struct CustomNonlinearityWithCache <: AbstractNonlinearPart end
2
3 function get_cache(::CustomNonlinearityWithCache, u0, p)
4     return (; cached_array=similar(u0))
5 end
```

[julia](#)

The cache content can be accessed in the main evaluation with `p.cache`:

```
1 function (nl::CustomNonlinearityWithCache)(du, u, p, t)
2     cached_array = p.cache.cached_array
3     ....
```

[julia](#)

Custom Nonlinearity: Raman example

Complete implementation in `../src/raman.jl`.

```
1  struct LLERamanNonlinearPart <: AbstractNonlinearPart end
2
3  # For Raman, we need to implement a cache to store intermediate computation
4  function get_cache(::LLERamanNonlinearPart, u0, p)
5      abs2_tmp = similar(u0) # pre-allocate an array for intermediate storage
6      raman_fourier_resp = convert(
7          typeof(u0),
8          @. -1im * 2π * p.freq * p.τ_R
9      ) # pre-computed the Rama kernel in spectral space
10     return (; abs2_tmp, raman_fourier_resp) # cache are supplied as a NamedTuple
11 end
```



```
1  @fastmath function (lle_raman::LLERamanNonlinearPart)(du, u, p, t) julia
2      S = get_var(lle_raman, p.S, p, t) # fetch eventually time varying function
3      Δ = get_var(lle_raman, p.Δ, p, t) # fetch eventually time varying function
4      @unpack abs2_tmp, raman_fourier_resp = p.cache # We can access the cache using
        p.cache
5      abs2_tmp .= abs2.(u) # used form temporary storage
6      @. du = 1im * (abs2_tmp - Δ) * u + S # LLE
7      fft!(p.bifft_plan, abs2_tmp) # in-place preallocated FFT
8      @. abs2_tmp = raman_fourier_resp * abs2_tmp # convolution = multiplication in
        spectral domain
9      ifft!(p.bifft_plan, abs2_tmp) # in-place preallocated IFFT
10     @. du += 1im * abs2_tmp * u
11     return du
12 end
```

To use `fft!` and `ifft!` with pre-allocated storage and temporary array, the wrappers `ifft!/fft!` (`p.bifft_plan, target`) are available.

Performance tips:

- Pre-allocate every array you may need in the cache (allocations are the main bottleneck of code)
 - Use Julia `similar(u0)` to create new array to keep to code generic.
 - Always use in-place mutating operation (in-place function in Julia are indicated with a!) which does not create temporary array
- Use the broadcast macro `@.` when to apply an element-wise operation on one or several arrays.
- Use the `@fastmath` macro before your function.
- Make sure all of your array have a well-defined type.
- Avoid using `abs(u)^2` to compute $|u|^2$, as it compute a squared-root internally. Use Julia `abs2()` instead
- Similarly, avoid `exp(1im*x)` to compute $\exp(ix)$ if x is real, use `cis(x)` instead.
- Be sure to computed only once and cached expensive computation that do not need to be updated.
- On IntelCPU, uses MKL FFT with `FFTW.set_provider!("mkl")`.
- Launch Julia with multiple threads wiht `julia --threads ...`
- For very large problem ($> 10^4$ points), consider using a GPU.
 - All the code in this repo should be compatible with a GPU by converting `u0` to `CuArray` from `CUDA.jl` package.
 - On the GPU, only use `Float32` (32-bit float) array instead of `Float64`.

Conclusion

- Except for simplicity, there is no reason to use Split-Step instead of RKIP
- Give Julia a try ! (You can interface it with Python)
- All examples code are on my GitHub <https://github.com/Azercoco/NLSE-Julia-Example>
 - Contributions and suggestions are welcome !